

Multi-Media Wizardry: How To Make ODS Outputs That “Dance and Sing”

LeRoy Bessler, Fox Point, Wisconsin, USA, bessler@execpc.com



Abstract

Augmenting simple, static, passive output should not dilute, obstruct, or distract from the message in and the meaning of your data. Some animation can be used to provide an essential data presentation function, rather than decoration or amusement. Audio can be a useful, extra, concurrent communication channel. And there are other multi-media options. For web pages, this paper shows you how to add audio (including background audio, which does not require launch of a media player on all or part of your screen), video, animation, a marquee ("traveling text", which is what you sometimes see at the bottom of a television news broadcast), and custom-sized images. For PDF documents, this paper shows you how to: include pop-up text (possibly data-dependent); create hyperlinks (possibly data-dependent) to audio clips, video clips, and animated graphics outside your PDF document; and imbed and re-size images. For RTF documents, there is no facility to add pop-up notes, and the handling of images and hyperlinks to external media files is analogous to that for other ODS output, but the distinguishing, special-effect possibility presented here is animated text. Though not really multi-media, it is important to see how the popular use of 3D graphics can impede, and even distort, communication. Acceptable use of 3D is demonstrated. The topics presented in this paper do not require the use of PROC TEMPLATE. This presentation is suitable for users with little or no ODS experience.

Introduction

My slide presentation can “dance and sing”, but this paper’s static collection of screen images cannot. Please feel free to email a request (to bessler@execpc.com) for a copy of the slides and supporting files to have the experience, or put the code below to work with your own data, and your own image, sound, and video files, and your own animated GIFs.

The Lively Web: Adding Sound and Motion to Your Web Pages (and Working with Images)



The web page above, when opened, plays “God Save the Queen”. The Queen’s personal flag (animated) is waving continuously. Both that image and the photo display pop-up text (a.k.a. HTML “ALT text”) when you rest your mouse on them. (When the pop-up window opens, note that a tool bar also appears near the upper left corner of the image.) One pop-up includes a line break—with multiple line breaks you can create a stack of multi-line pop-up text as high as you need/want, even to the extent of displaying a simple table. Software for visually impaired web users can read your HTML file, including the ALT text description of images, to convert that text to audio. **LINK=**, used to provide the TITLE hyperlink, can also be used in a FOOTNOTE. An explanation follows this listing of the code:

```
data _null_;
text = "'Queen & Prince board airplane" || '0D'X || "Click for the British Monarchy web
site'";
/* '0D'X forces the line break */
call symput('AltText',text); run;

ods html path='C:\demo\WebPages' (url=None) file='Aug2005.html' style=styles.minimal;
title color=CX0000FF font='Georgia' height=20 pt
link='http://www.views-uk.org' 'Go To VIEWS Home Page';
proc print label noobs data=sashelp.class(where=(name EQ 'Philip'))
style(header) = [font_face=Georgia font_size=6 cellwidth=300]
style(data) = [foreground=CXFFFFFF cellheight=1 cellwidth=1]
style = [rules=None cellspacing=0 frame=void borderwidth=0 cellpadding=0]
prehtml = "<img src='../Images\WavingFlagOfQueen.gif' width=89 height=101
alt='Waving Flag of Her Majesty The Queen'>
<font face=Georgia size=7 color=#FFFF00>
<marquee bgcolor=#000099 width=100% hspace=20 vspace=10>
God save our gracious Queen. Long live our noble Queen. God save the Queen! Send her
victorious, happy, and glorious, long to reign over us. God save the Queen!
</marquee>
</font>
<a href='http://www.royal.gov.uk'>
<img src='../Images\Boarding.jpg' vspace=10 width=300 height=225 alt=&AltText>
</a>
<bgsound loop=infinite src='../Audios\UK_anthem.wav'>"]";
var name; label name='Returning Home'; run;
ods html close;
```

Instead of relying on PROC TEMPLATE and a custom style, the ODS controls are supplied via the PROC PRINT statement. Here, for demonstration, the PROC PRINT presents an image caption (provided by the LABEL statement) via column header cell content. The data cell is shrunk and hidden. A real application, rather than this all-in-one proof of concept, would include a real data table and/or graph, and would use only audio and visual features that suit the application’s total communication goal.

The HTML code used here is delivered via ODS using the PREHTML parameter. One can also or instead use a POSTHTML parameter to deliver HTML code that is applied after the tabular (or graphic) PROC output in the web page. E.g., a marquee and/or image(s) can be located below the PROC output.

The BGSOUND tag causes the web browser PC to play the sound file without opening a media player window, unlike what happens when you click on a hyperlink to a sound file. You can set LOOP to any integer you like, rather than to INFINITE.

HTML colors are RGB colors. You can use SAS® RGB color values, but with # replacing the CX prefix. BGCOLOR in the MARQUEE tag is the color of the marquee background.

The MARQUEE tag “packages” the traveling text, but formatting the text must be done with the “surrounding” FONT tag. HTML font sizes are integers from 1 to 7. (ODS converts any point sizes in your SAS code to HTML font sizes.)


WIDTH, HEIGHT, HSPACE, and VSPACE HTML values are in pixels, except that a percent assignment for WIDTH is percent of web page width. VSPACE is space above *and* below the marquee or image. HSPACE is space at the left *and* right. Requesting the insertion of space only above, only below, only to the left, or only to the right is not supported in HTML.

SRC= identifies the location of the image or sound file. Both the animated GIF and the static JPG file are resized, but with care to preserve the aspect ratio (i.e., the ratio of picture height to picture width) to avoid distortion. (The Queen’s flag is a square as shown, not the rectangular shape that is usually expected for flags.)

The ANCHOR tag (<a . . . >) with HREF= makes the photo image a hyperlink. (Animated images can be hyperlinks, too.)

Animation for Communication, Not Decoration or Recreation

In the 2D world of the screen, time can be handled as a third dimension, by serially presenting two-dimensional graphs using animation. The graphs can be any type: pie chart, bar chart, trend line, scatter plot, map, etc. (For an example of a map, which includes annotation of the name and changing population of each of its geographic unit areas, see pages 30-31 in Reference 2.) Below are three (shrunk) panels from an animated graph showing the growth of weight with age. (To save publication space, the web browser and Windows framing have been clipped away from the screen images, and some panels had to be omitted.)



Here is the code used:

```
proc catalog cat=work.gseg kill; run; quit; /* clear out any leftovers from a prior code run during the SAS session */
```

```
/* The use of custom XPIXELS and YPIXELS in sizing below requires creation of a modified graph device driver.
```

```
For more information, see SAS Note SN-005366 at support.sas.com. */
```

```
libname gdevice0 "C:\AnyFolder";
```

```
proc gdevice nofs cat=gdevice0.devices;
```

```
delete modanim;
```

```
copy gifanim from=sashelp.devices newname=modanim;
```

```
modify modanim hsize=0 vsize=0; run; quit;
```

```
/* This section of code needed only to web publish the AnimDemo.GIF file. */
```

```
ods noresults; ods listing close;
```

```
ods html path="c:\Folder" (url=none) file="WeightByAge.html" (title="Animated GIF of Student Weight By Age")
```

```
gtitle gfootnote style=styles.YourFavoriteCustomOrODSprovidedStyle;
```

```
GOPTIONS RESET=ALL DEVICE=GIF XPIXELS=994 YPIXELS=555; /* Size the image to take maximum advantage of the  
web browser window size, but avoid any need for scrolling. The pixel counts assume that screen resolution is 1024 X 768, that  
the web browser is Microsoft Internet Explorer 6.0, and that no extra tool bars—such as that for Yahoo—are user enabled. */
```

```
PROC GSLIDE NAME="AnimDemo";
```

```
TITLE1 'this mandatory static placeholder image will never actually display'; RUN; QUIT;
```

```
ods html close; ods listing;
```

```
proc sort data=sashelp.class out=work.ToChart;
```

```
by Age; run;
```

```
proc means data=sashelp.class noprint max;
```

```
var Weight;
```

```
output out=FoundMax Max=MaximumWeight; run;
```

```
data _null_;
```

```
set FoundMax;
```

```
call symput('MaxWeight',trim(left(MaximumWeight))); run;
```

```
GOPTIONS RESET=ALL DEVICE=modanim
```

```
ITERATION=3 /* show the sequence of animation panels 3 extra times after the first time */
```

```
DELAY=100 /* display time for each animation panel in units of 0.01 seconds, but actual time is display-tool-dependent */
```

```
XPIXELS=994 YPIXELS=555 /* See sizing comment above. If not web publishing, use XPIXELS=800 YPIXELS=600. */
```

```
GSFNAME=animout GSFMODE=replace CBACK=CXFFFF00 HTEXT=32 PT FTEXT="Georgia/Bold" CTEXT=CX000000;
```

```
TITLE1 H=32 PT F="Georgia/Bold" C=CX000000 "Weight of Students Who Are Age " C=CX9900CC "#BYVAL(Age)"
```

```
J=C C=CX0000FF "Maximum Weight of Any Student is &MaxWeight";
```

```
PATTERN1 V=SOLID C=CX00CC00;
```

```
AXIS1 LABEL=NONE MAJOR=NONE MINOR=NONE STYLE=0 VALUE=(J=R);
```

```
AXIS2 LABEL=NONE MAJOR=NONE MINOR=NONE STYLE=0
```

```
ORDER=(0 &MaxWeight) VALUE=(TICK=1 '0' TICK=2 C=CX0000FF "&MaxWeight") OFFSET=(+3,+7);
```

```
OPTIONS NOBYLINE;
```

```

FILENAME animout "c:\Folder\AnimDemo.gif";
PROC GCHART DATA = work.ToChart;
BY Age;
HBAR Name / SUMVAR=Weight SUMLABEL="Weight" WIDTH=2 SPACE=2 MAXIS=axis1 RAXIS=axis2
HREF=&MaxWeight LREF=33;
RUN; QUIT;
FILENAME animout CLEAR;

```

An animated GIF file can be displayed outside of html. E.g., you can insert it as a picture in a Microsoft PowerPoint slide. For that target application, you can use the same code as above except that you: (a) must change XPIXELS and YPIXELS to 800 and 600 in order to make best use of the slide display area; and (b) may omit the web publishing code section listed in the box.

Action In Or From PDF

Pop-Up Text for Your PDF Document

If you would like some information as flyover/pop-up text, not hardcopy, in your PDF document, you can create a (scrollable) note box “behind” an icon. The icon does not print, and the note does not print. You can right-click the icon to open the note box. There is no apparent limit on the length of note text. You can define the width of the note box. If you take the default box width, the icon will overlay the left of the “hard text” on the page. Increasing its height would push the table around.


With this code:

```


options reset=all;
options orientation=portrait;
ods pdf notoc startpage=never file="&FolderForPDF.\TitleWithFlyoverText.pdf";
ods pdf text="^S={FLYOVER='This is the CLASS data set from the SASHELP data library'
                CELLWIDTH=6IN JUST=CENTER FONT_SIZE=16PT FONT_FACE=Helvetica}Students";
proc print data=sashelp.class; run;
ods pdf close;

```


you get this result (in a series of clippings from the PDF page), starting with display of the icon:



Then resting the mouse on the icon:



And finally opening the note:




Data-Dependent Pop-Up Text in Table Cells

Use a SAS format to selectively define the pop-up text based on data value, with code such as this:

```
proc format lib=work;
value AgeNote
    11 = 'Youngest'
    16 = 'Oldest'
    Other = ' ';
run;

options reset=all;
options orientation=portrait;
ods pdf notoc file="&FolderForPDF.\SomeDataCellsWithFlyoverText.pdf";
proc print data=sashelp.class;
var Name Sex;
var Age / style=[CELLWIDTH=1.2IN /* widen the cells to prevent the notes
                                from overlaying the data */
                just=right flyover=AgeNote.] ;
var Height Weight;
run;
ods pdf close;
```

Here is a clipping from the PDF page with one of the data-dependent pop-up notes opened:



10	John	M	12	59.0	99.5
11	Joyce	F	11	51.3	50.5
12	Judy	F	14	64.3	90.0
13	Louise	F	12	56.3	77.0
14	Mary	F	15	66.5	112.0
15	Philip	M	16	72.0	150.0
16	Robert	M	12	64.8	128.0
17	Ronald	M	15	67.0	133.0
18	Thomas	M	11	57.5	85.0
19	William	M	15	66.5	112.0


Using an Image As a Hyperlink in Your PDF Document

PREIMAGE and/or POSTIMAGE and/or BACKGROUNDIMAGE options can be used to display an image (or multiple images) anywhere in tabular report output from a SAS PROC. The URL option can be used in TITLE, FOOTNOTE, or PDF TEXT statements, or in cells for data or for column headers, to create the hyperlink(s). You can, of course, use a graph as a hyperlink, if you created it as an image file (e.g., with the GIF device driver).

Omitting the ODS code and the TITLE statements, code that can be used to display an image hyperlink is:

```
proc print data=sashelp.class
    (where=(name='Alfred')) noobs label
    style(table) =[frame=void]
    style(header)=[url='http://www.sas.com'
                    backgroundimage="&Path.\images\photo1.jpg"
                    background=white
                    cellwidth=12 cm cellheight=9 cm] /* magnify,
                                                    but preserve the 4 X 3 aspect
ratio */
    style(data)  =[foreground=white
                    font_size=1 pt];
var age;
label age='00'X; run;
```

The PROC PRINT creates an invisible table as the frame for the image/hyperlink. Here is the result:



Putting It All Together: Image, Audio, and Video for a Multi-media PDF Extravaganza

I collected elements for this example PDF document as follows. At <http://www.phuse.info/photos/photo1.html>, I right-clicked on the image, then selected "Save Picture As", and stored the image as photo1.jpg in C:\PDF\images. I downloaded the Bavarian Anthem sound file from david.national-anthems.net/bav.htm, and stored it as bav.wav in C:\PDF\audios. Using Microsoft Windows Explorer, I searched for PC-resident files of the form *.avi. I selected one from the Program Files for SAS (in C:\Program Files\SAS Institute\SAS\V8\core\sasmisc) and stored it as tiger.avi in C:\PDF\videos.

```
%let urlPath = %str(C:/PDF);
%let filePath = %str(C:\PDF);
%let commonfmt = %str(JUST=CENTER VJUST=MIDDLE FONT_SIZE=12PT CELLHEIGHT=24PT);
%let TOCfmt = &commonfmt%str( CELLWIDTH=10CM);
%let TXTfmt = &commonfmt%str( CELLWIDTH=20CM);

proc format;
value $AVlnks
'Inside Stadthalle' = "&urlPath./images/photo6.jpg"
'Anthem of Bavaria' = "&urlPath./audios/bav.wav"
'SAS Video' = "&urlPath./videos/tiger.avi";
run;

data work.ChoicesForImageAudioVideo;
length medium $ 17;
medium = 'Inside Stadthalle'; output;
medium = 'Anthem of Bavaria'; output;
medium = 'SAS Video'; output;
run;
```

```

options nodate nonumber orientation=landscape;
ods escapechar="^";
ods noresults;
ods listing close;
ods pdf notoc file=" . . . " startpage=never;


title1 . . . ; title2 . . . ;
ods pdf text="^S={&TOCfmt URL="&urlPath./images/CosmoKramer.jpg"}Click . . . ";
ods pdf text="^S={&TOCfmt URL="&urlPath./audios/brute.wav"} . . . ";
ods pdf text="^S={&TXTfmt}";
ods pdf text="^S={&TXTfmt} . . . ";

proc print data=work.ChoicesForImageAudioVideo noobs label
  style=[preimage="&filePath.\images\photo1.jpg" frame=void];
var medium /
  style(data) =[URL=$AVlnks. font_size=12pt
    cellwidth=4.5 cm just=center cellheight=0.8 cm vjust=middle]
  style(header)=[background=white font_size=12pt font_weight=medium
    just=center cellheight=0.8 cm vjust=middle];
label Medium='Click to launch file: ';
run;

ods pdf close;
ods listing;

```

Here is the result for the “home page”, but no illustrations of the screen images of the linked-to destinations:




Animated RTF Text: Some of These Options May Confuse the Possible With the Useful and Appropriate

You can use ODS to create an RTF file with imbedded tables, graphs (but animated GIFs do NOT work in RTF documents), images, and hyperlinks to external audio or video files. Since the coding for such applications is analogous to that in the prior section on PDF, none will be presented here. There is no support for user-created pop-up text in RTF (even though Microsoft Word itself does use pop-up text in the document window for certain communication with the user—such as when your mouse rests on a hyperlink in a Word document).

Apart from drawing attention to text without use of bold or underline or large print, the value and purpose of animated text as implemented in RTF is questionable. Just as blinking text in a web page obstructs the effectiveness of software used to convert web page content into audio for visually impaired users, animated RTF text may impose a usability problem for certain users. I show how to create animated RTF text on the presumption that anyone who implements it will address the usability question.

Here is a clipped out and enlarged example of all six types of RTF text animation. What you see below is only a pasted-in static screen print. Though it would be better for you to just run the code and see the result yourself, let me try to describe the animated effects. In Effect 1, the rectangular border of symbols switches from green X's to blue solid squares to blue-green diamonds to green X's in a continuous cycle. In Effect 2, the black overlay appears and disappears on top of the “Peek-A-Boo” text. In Effect 3, the multi-colored symbols “twinkle”, but intermittently all of them vanish so that you can briefly see all of the text clearly. In Effects 4 and 5, the black and red dashes travel around the rectangular track. In Effect 6, the text alternates between a solid font and the fuzzy font that you see in the screen print. When the text is printed, none of these effects are present. In my opinion, Effect 5 is conspicuous, without being irritating or obstructive. (But the rotating border does obscure the bottom of the letter “g”.)



The `\ANIMTEXTn` control word (also available for FOOTNOTE statements) is used in this code to create the animated text:


```
ods rtf body='c:\RTFdemo\ProbablyFrivolous.rtf' style=styles.minimal;
title1 h=16pt '{\animtext1 This message shows animation effect 1.}';
title2 h=16pt '{\animtext2 This message shows animation effect 2.}';
title3 h=16pt '{\animtext3 This message shows animation effect 3.}';
title4 h=16pt '{\animtext4 This message shows animation effect 4.}';
title5 h=16pt '{\animtext5 This message shows animation effect 5.}';
title6 h=16pt '{\animtext6 This message shows animation effect 6.}';
proc print data=sashelp.class; run;
ods rtf close;
```

When and How to Use 3D

Strictly speaking, 3D is not multi-media, but this “multiplication by 1.5” attempts to stretch the power of a 2D medium, be it the printed page or the screen. 3D is the commonest tool for attempts to “enhance” graphics. I recommend its use for presentation of three variables, not for adding needless complexity (and possibly distortion) to the presentation of two variables.


The most frequently used tools for adding “sizzle” to a “boring” 2D graphic presentation of data are the 3D pie chart and the 3D bar chart. I cannot provide much code here, but simply present some finished 3D charts, and, if needed, my recommended 2D alternative. (Code—for the recommended charts—can be supplied upon request. My email address is at the end of the paper.)

The 3D Pie Chart Is Very Commonly Used **BUT**
Always Distorts the Apparent Relative Size of Slices.




5% slice at rear is twice as big as 5% slice at right
11% slice is almost as big as 19% slice

2D Pie Chart. **Simpler is Better.**
Always Presents Accurate Relative Size of Slices.




Can anyone who has seen the chart above (which was based on real data in a real application developed by me early in my graphic self-education) **ever want to create a 3D pie chart?** The case against the 3D pie chart is based upon its distortion, upon its anti-communication. However, the 3D bar chart introduces no distortion, just needless complication. Here is another face-off:

3D Vertical Bar Chart. **Difficult to Use.**



2D Vertical Bar Chart.


Easy Visual Comparison AND Precise Values Obvious



When I feel the urge to “fancify” a bar chart, I favor the use of narrow diameter cylinders in a horizontal bar chart like these:


3D Horizontal Bar Chart.

Interest Added, But Still Communication-Effective



Useful When a Pie Chart Is Infeasible:

Horizontal Bar Chart Can List Percent of Total



Here is the essential code (excluding, e.g., GOPTIONS) for the above pie chart alternative:

```
pattern1 c=CX00FF00;
axis1 label=none major=none minor=none style=0 noplane;
axis2 label=none major=none minor=none style=0 noplane value=none;
filename anyname "C:\3DPieChartAlternative.png";
proc gchart data=DataForSimpleCharts;
hbar3d Name / freq=Value freq freqlabel='Sales in Billions'
  /* Value is NOT a frequency, but this trick makes the graph possible and correct */
  percent percentlabel='Percent of Market'
  descending noframe shape=cylinder width=0.6 space=0.6 maxis=axis1 raxis=axis2;
run; quit;
```

Conclusion

You can do some fancy things with ODS, a little bit of HTML, and, optionally, SAS/GRAPH. Some of the examples were devoid of real data. That was out of expediency—I wanted the result, and the way to get there, to focus on the particular active elements being demonstrated. I am confident that the user can apply this multi-media wizardry to real data presentation contexts. These are methods that have worked for me, and I am always anxious to learn new ones. I would be happy to hear from you about yours.

Annotated Bibliography of Related Work by This Author

1. “Adding Sound and Motion to Your Web Pages and Working with Images”, *VIEWS News: News, Hints, Tips, and Information for SAS Users*, Issue 31, 3rd Quarter 2005, Views International SAS Programmer Community (United Kingdom), available at <http://www.views-uk.org>. This is a prior publication of what I present above in my section on the “The Lively Web”.
2. With Francesca Pierri, “Your Graphs and Tables at Their Best on the Web with ODS” in *Proceedings of the SAS Users Chicago International Conference 2001* (Fox Point, Wisconsin, USA), and *VIEWS 2001 Conference Proceedings*, United Kingdom SAS Users Group (London, UK), 2001. This expansive paper (31 pages) is available from LeRB via email.
3. “Getting Started with, and Getting the Most out of, SAS ODS PDF: No Mastery of PROC TEMPLATE Required”, *Proceedings of the PhUSE 2005 Conference*, Pharmaceutical Users Software Exchange (Heidelberg, Germany), 2005. Find it on the web at <http://www.lexjansen.com/phuse/2005/ts/ts05.pdf>. The paper extends some work that I did with Deborah Skinner.
4. “Easy, Elegant, and Effective SAS Graphs: Inform and Influence with Your Data”, *Proceedings of the PhUSE 2005 Conference*, Pharmaceutical Users Software Exchange (Heidelberg, Germany), 2005. Find it at on the web at <http://www.lexjansen.com/phuse/2005/tu/tu07.pdf>.
5. “3D versus 2D Presentation Graphics”, *VIEWS News: News, Hints, Tips, and Information for SAS Users*, Issue 29, 1st Quarter 2005, Views International SAS Programmer Community (United Kingdom), available at <http://www.views-uk.org>.
6. “Web Communication Effectiveness: Design and Methods to Get the Best Out of ODS, SAS, and SAS/GRAPH”, *Proceedings of the Twenty-Eighth Annual SAS Users Group International Conference*, 2003. Cary, NC: SAS Institute Inc.

Acknowledgments

I am grateful to John Xu and Delayne Stokke, the MWSUG 2007 Conference Co-chairs, for providing me an opportunity to share this work with other SAS users.

Notices

SAS is a registered trademark or trademark of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other product and brand names are trademarks or registered trademarks of their respective owners.

Author Information

Your questions, your comments, and your guidance as to how to perform other feats of multi-media wizardry are welcome.

LeRoy Bessler PhD, bessler@execpc.com

Zum Sehen geboren, zum Schauen bestellt
Born to See, Meant to Look
— Goethe

Ancoro imparo
I am still learning
— Michelangelo, at age 87